

20 *Exact Belief State Planning*

Because states are not directly observable, the agent must use its past history of actions and observations to inform its belief, which can be represented as a probability distribution over states as discussed in the previous chapter. The objective in a POMDP is to choose actions that maximize the accumulation of reward while interacting with the environment. There are different approaches for computing an optimal policy that maps beliefs to actions given models of the transitions, observations, and rewards.¹ One approach is to convert a POMDP into an MDP and apply dynamic programming. Other approaches include representing policies as conditional plans or as piecewise linear value functions over the belief space. The chapter concludes with an algorithm for computing an optimal policy that is analogous to value iteration for MDPs.

¹ A discussion of exact solution methods is provided by L. P. Kaelbling, M. L. Littman, and A. R. Cassandra, "Planning and Acting in Partially Observable Stochastic Domains," *Artificial Intelligence*, vol. 101, no. 1-2, pp. 99-134, 1998.

20.1 *Belief-State Markov Decision Processes*

Any POMDP can be viewed as an MDP that uses beliefs as states, also called a *belief-state MDP*.² The state space of a belief-state MDP is the set of all beliefs \mathcal{B} . The action space is identical to that of the POMDP. If the state and observation spaces are discrete, the belief-state transition function for a belief-state MDP is

² K. J. Åström, "Optimal Control of Markov Processes with Incomplete State Information," *Journal of Mathematical Analysis and Applications*, vol. 10, no. 1, pp. 174-205, 1965.

given by:

$$T(b' | b, a) = P(b' | b, a) \quad (20.1)$$

$$= \sum_o P(b' | b, a, o) P(o | b, a) \quad (20.2)$$

$$= \sum_o P(b' | b, a, o) \sum_s P(o | b, a, s) P(s | b, a) \quad (20.3)$$

$$= \sum_o P(b' | b, a, o) \sum_s P(o | b, a, s) b(s) \quad (20.4)$$

$$= \sum_o P(b' | b, a, o) \sum_{s'} \sum_s P(o | b, a, s, s') P(s' | b, s, a) b(s) \quad (20.5)$$

$$= \sum_o (b' = \text{Update}(b, a, o)) \sum_{s'} P(o | a, s') \sum_s T(s' | s, a) b(s) \quad (20.6)$$

Above, $\text{Update}(b, a, o)$ returns the updated belief using the deterministic process discussed in the previous chapter.³ For continuous problems, we replace the summations with integrals.

The reward function for a belief-state MDP depends on the belief and action taken. It is simply the expected value of the reward:

$$R(b, a) = \sum_s R(s, a) b(s) \quad (20.7)$$

Solving belief-state MDPs is challenging because the state space is continuous. We can use the approximate dynamic programming techniques presented in earlier chapters, but we can often do better by taking advantage of the structure of the belief-state MDP, as will be discussed in the remainder of this chapter.

20.2 Conditional Plans

There are different ways to represent policies for POMDPs. One approach is to use a *conditional plan* represented as a tree. Figure 20.1 shows an example of a three-step conditional plan with binary action and observation spaces. The nodes correspond to belief states. The edges are annotated with observations, and the nodes are annotated with actions. If we have a plan π , the action associated with the root is denoted $\pi()$ and the subplan associated with observation o is denoted $\pi(o)$. Algorithm 20.1 provides an implementation.

A conditional plan tells us what to do in response to our observations up to the horizon represented by the tree. To execute a conditional plan, we start with




Figure 20.1. An example 3-step conditional plan.

³ As a reminder, we use the convention where a logical statement in parentheses is treated numerically as 1 when true and 0 when false.

```

struct ConditionalPlan
    a          # action to take at root
    subplans   # dictionary mapping observations to subplans
end

ConditionalPlan(a) = ConditionalPlan(a, Dict{})

(π::ConditionalPlan)() = π.a
(π::ConditionalPlan)(o) = π.subplans[o]

```

Algorithm 20.1. The conditional plan data structure consisting of an action and a mapping from observations to subplans. The `subplans` field is a `Dict` from observations to conditional plans. For convenience, we have created a special constructor for plans consisting of a single node.

the root node and execute the action associated with it. We proceed down the tree according to our observations, taking the actions associated with the nodes through which we pass.

Suppose we have a conditional plan π , and we want to compute its expected utility when starting from state s . This computation can be done recursively:

$$U^\pi(s) = R(s, \pi()) + \gamma \left[\sum_{s'} T(s' | s, \pi()) \sum_o O(o | \pi(), s') U^{\pi(o)}(s') \right] \quad (20.8)$$

An implementation for this procedure is given in algorithm 20.2.

```

function lookahead(ℙ::POMDP, U, s, a)
    S, O, T, O, R, γ = ℙ.S, ℙ.O, ℙ.T, ℙ.O, ℙ.R, ℙ.γ
    u' = sum(T(s,a,s')*sum(O(a,s',o)*U(o,s') for o in O) for s' in S)
    return R(s,a) + γ*u'
end

function evaluate_plan(ℙ::POMDP, π::ConditionalPlan, s)
    U(o,s') = evaluate_plan(ℙ, π(o), s')
    return isempty(π.subplans) ? ℙ.R(s,π()) : lookahead(ℙ, U, s, π())
end

```

Algorithm 20.2. A method for evaluating a conditional plan π for MDP \mathcal{P} starting at state s . Plans are represented as tuples consisting of an action and a dictionary mapping observations to subplans.

If we do not know the current state exactly, we can compute the utility of our belief b as follows:


$$U^\pi(b) = \sum_s b(s) U^\pi(s) \quad (20.9)$$

Example 20.1 shows how to compute the utility associated with a three-step conditional plan.

Now that we have a way to evaluate conditional plans up to some horizon h , we can compute the optimal h -step value function:


$$U^*(b) = \max_{\pi} U^\pi(b) \quad (20.10)$$

Consider the following 3-step conditional plan for the crying baby problem:



In this plan, we begin by ignoring the baby. If we observe any crying, we feed the baby. If we do not observe any crying, we ignore the baby. Our third action again feeds if there is crying.

The expected utility for this plan in belief space is plotted alongside a 3-step plan that always feeds the baby and one that always ignores the baby.



We find that the given plan is not universally better than either always ignoring or always feeding the baby.

Example 20.1. A conditional plan for the three-step crying baby problem (appendix F.7), evaluated and compared to two simpler conditional plans.

An optimal action can be generated from the action associated with the root of a maximizing π .

Solving an h -step POMDP by directly enumerating all h -step conditional plans is generally computationally intractable as shown in figure 20.2. There are $(|\mathcal{O}|^h - 1)/(|\mathcal{O}| - 1)$ nodes in an h -step plan. In general, any action can be inserted into any node, resulting in $|\mathcal{A}|^{(|\mathcal{O}|^h - 1)/(|\mathcal{O}| - 1)}$ possible h -step plans. This exponential growth means enumerating over all plans is intractable even for modest values of h . As will be discussed later in this chapter, there are alternatives to explicitly enumerating over all possible plans.

20.3 Alpha Vectors

We can rewrite equation (20.9) in vector form:

$$U^\pi(b) = \sum_s b(s) U^\pi(s) = \alpha_\pi^\top \mathbf{b} \quad (20.11)$$

The vector α_π is called an *alpha vector* and contains the expected utility under plan π for each state. As with belief vectors, alpha vectors have dimension $|S|$. Unlike beliefs, the components in alpha vectors represent utilities and not probability masses. Algorithm 20.3 shows how to compute an alpha vector.

```
function alphavector( $\mathcal{P}$ ::POMDP,  $\pi$ ::ConditionalPlan)
    return [evaluate_plan( $\mathcal{P}$ ,  $\pi$ ,  $s$ ) for  $s$  in  $\mathcal{P}.S$ ]
end
```

Each alpha vector defines a hyperplane in belief space. The optimal value function equation (20.11) is the maximum over these hyperplanes,

$$U^*(\mathbf{b}) = \max_{\pi} \alpha_\pi^\top \mathbf{b} \quad (20.12)$$

making the value function piecewise-linear and convex.⁴

An alternative to using a conditional plan to represent a policy is to use a set of alpha vectors Γ , each annotated with an action. Although not practical, one way to generate the set Γ is to enumerate the set of h -step conditional plans and compute their alpha vectors. The action associated with an alpha vector is the action at the root of the associated conditional plan. We execute a policy represented by Γ by updating our belief state and executing the action associated with the dominating




Figure 20.2. Even for extremely simple POMDPs such as the crying baby problem with three actions and two observations, the number of plans grows extremely quickly with the planning horizon. In comparison, only 2 plans dominate in each step for the crying baby problem, so only 8 plans need be evaluated with each iteration: the 2 actions as roots and the 2 dominant plans as subtrees for each observation.

Algorithm 20.3. We can generate an alpha vector from a conditional plan by calling `evaluate_plan` from all possible initial states.

⁴ The optimal value function for continuous-state POMDPs is also convex, as can be seen by approximating the POMDP through state space discretization and taking the limit as the number of discrete states approaches infinity.

alpha vector at the new belief \mathbf{b} . The dominating alpha vector α at \mathbf{b} is the one that maximizes $\alpha^\top \mathbf{b}$. This strategy can be used to select actions beyond the horizon of the original conditional plans. Algorithm 20.4 provides an implementation.

```

struct AlphaVectorPolicy
   $\mathcal{P}$  # POMDP problem
   $\Gamma$  # alpha vectors
   $\mathbf{a}$  # actions associated with alpha vectors
end

function utility( $\pi$ ::AlphaVectorPolicy,  $\mathbf{b}$ )
  return maximum( $\alpha \cdot \mathbf{b}$  for  $\alpha$  in  $\pi.\Gamma$ )
end

function ( $\pi$ ::AlphaVectorPolicy)( $\mathbf{b}$ )
   $\mathbf{u}, \mathbf{i}$  = findmax( $[\alpha \cdot \mathbf{b}$  for  $\alpha$  in  $\pi.\Gamma$ ])
  return ( $\mathbf{a}=\pi.\mathbf{a}[\mathbf{i}]$ ,  $\mathbf{u}=\mathbf{u}$ )
end

```

Algorithm 20.4. An alpha vector policy is defined in terms of a set of alpha vectors Γ and an array of associated actions \mathbf{a} . Given current belief \mathbf{b} , it will find the alpha vector that gives the highest value at that belief point. It will return the associated action and its utility.

If we use *one-step lookahead*, we do not have to keep track of the actions associated with the alpha vectors in Γ . The one-step lookahead action from belief \mathbf{b} using the value function represented by Γ , denoted U^Γ , is

$$\pi^\Gamma(\mathbf{b}) = \arg \max_a \left[R(\mathbf{b}, a) + \gamma \sum_o P(o \mid \mathbf{b}, a) U^\Gamma(\text{Update}(\mathbf{b}, a, o)) \right] \quad (20.13)$$

where

$$P(o \mid \mathbf{b}, a) = \sum_s P(o \mid s, a) b(s) \quad (20.14)$$

$$P(o \mid s, a) = \sum_{s'} T(s' \mid s, a) O(o \mid s', a) \quad (20.15)$$

Algorithm 20.5 provides an implementation. Example 20.2 demonstrates one-step lookahead on the crying baby problem.

20.4 Pruning

If we have a collection of alpha vectors Γ , we may want to *prune* alpha vectors that do not contribute to our representation of the value function or plans that are not optimal for any belief. Removing such alpha vectors or plans can improve computational efficiency. We can check whether an alpha vector α is *dominated* by

```

function lookahead( $\mathcal{P}$ ::POMDP, U, b::Vector, a)
     $S, \mathcal{O}, T, \mathcal{O}, R, \gamma = \mathcal{P}.S, \mathcal{P}.\mathcal{O}, \mathcal{P}.T, \mathcal{P}.\mathcal{O}, \mathcal{P}.R, \mathcal{P}.\gamma$ 
     $r = \text{sum}(R(s,a)*b[i] \text{ for } (i,s) \text{ in enumerate}(S))$ 
     $\text{Posa}(o,s,a) = \text{sum}(\mathcal{O}(a,s',o)*T(s,a,s') \text{ for } s' \text{ in } S)$ 
     $\text{Poba}(o,b,a) = \text{sum}(b[i]*\text{Posa}(o,s,a) \text{ for } (i,s) \text{ in enumerate}(S))$ 
    return  $r + \gamma*\text{sum}(\text{Poba}(o,b,a)*U(\text{update}(b, \mathcal{P}, a, o)) \text{ for } o \text{ in } \mathcal{O})$ 
end

function greedy( $\mathcal{P}$ ::POMDP, U, b::Vector)
    u, a = _findmax(a  $\rightarrow$  lookahead( $\mathcal{P}$ , U, b, a),  $\mathcal{P}.\mathcal{A}$ )
    return (a=a, u=u)
end

struct LookaheadAlphaVectorPolicy
     $\mathcal{P}$  # POMDP problem
     $\Gamma$  # alpha vectors
end

function utility( $\pi$ ::LookaheadAlphaVectorPolicy, b)
    return maximum( $\alpha \cdot b$  for  $\alpha$  in  $\pi.\Gamma$ )
end

function ( $\pi$ ::LookaheadAlphaVectorPolicy)(b)
    U(b) = utility( $\pi$ , b)
    return greedy( $\pi.\mathcal{P}$ , U, b)
end

```

Algorithm 20.5. A policy represented by a set of alpha vectors Γ . It uses one-step lookahead to produce an optimal action and associated utility. Equation (20.13) is used to compute the lookahead.

Consider using one-step lookahead on the crying baby problem with a value function given by the alpha vectors $[-3.7, -15]$ and $[-2, -21]$. Suppose our current belief is $b = [0.5, 0.5]$, meaning that we believe it is equally likely the baby is hungry as not hungry. We apply equation (20.13):

$$\begin{aligned}
 & R(b, \text{feed}) = -10 \\
 & \quad \gamma P(\text{crying} \mid b, \text{feed}) U(\text{Update}(b, \text{feed}, \text{crying})) = -0.18 \\
 & \quad \gamma P(\text{quiet} \mid b, \text{feed}) U(\text{Update}(b, \text{feed}, \text{quiet})) = -1.62 \\
 & \quad \rightarrow Q(b, \text{feed}) = -11.8 \\
 & R(b, \text{ignore}) = -5 \\
 & \quad \gamma P(\text{crying} \mid b, \text{ignore}) U(\text{Update}(b, \text{ignore}, \text{crying})) = -6.09 \\
 & \quad \gamma P(\text{quiet} \mid b, \text{ignore}) U(\text{Update}(b, \text{ignore}, \text{quiet})) = -2.81 \\
 & \quad \rightarrow Q(b, \text{ignore}) = -13.9 \\
 & R(b, \text{sing}) = -5.5 \\
 & \quad \gamma P(\text{crying} \mid b, \text{sing}) U(\text{Update}(b, \text{sing}, \text{crying})) = -6.09 \\
 & \quad \gamma P(\text{quiet} \mid b, \text{sing}) U(\text{Update}(b, \text{sing}, \text{quiet})) = -1.85 \\
 & \quad \rightarrow Q(b, \text{sing}) = -14.0
 \end{aligned}$$

We use $Q(b, a)$ to represent the action value function from a belief state. The policy predicts that feeding the baby will result in the highest expected utility, so it takes that action.

Example 20.2. An example of a lookahead policy applied to the crying baby problem.

the alpha vectors in a set Γ by solving a linear program to maximize the utility gap δ that vector achieves over all other vectors:⁵

$$\begin{aligned}
 & \underset{\delta, \mathbf{b}}{\text{maximize}} && \delta \\
 & \text{subject to} && \mathbf{b} \geq \mathbf{0} \\
 & && \mathbf{1}^\top \mathbf{b} = 1 \\
 & && \boldsymbol{\alpha}^\top \mathbf{b} \geq \boldsymbol{\alpha}'^\top \mathbf{b} + \delta, \quad \boldsymbol{\alpha}' \in \Gamma
 \end{aligned} \tag{20.16}$$

⁵ Constraints of the form $\mathbf{a} \geq \mathbf{b}$ are element-wise. That is, we mean $a_i \geq b_i$ for all i .

The first two constraints ensure that \mathbf{b} is a categorical distribution, and the final set of constraints ensures that we find a belief vector for which $\boldsymbol{\alpha}$ has a higher expected reward than all alpha vectors in Γ . If, after solving the linear program, the utility gap δ is negative, then $\boldsymbol{\alpha}$ is dominated. If δ is positive, then $\boldsymbol{\alpha}$ is not dominated and \mathbf{b} is a belief at which $\boldsymbol{\alpha}$ is not dominated. Algorithm 20.6 provides an implementation for solving equation (20.16) to determine a belief, if one exists, where δ is most positive.

```

function find_maximal_belief( $\alpha$ ,  $\Gamma$ )
    m = length( $\alpha$ )
    if isempty( $\Gamma$ )
        return fill(1/m, m) # arbitrary belief
    end
    model = Model{GLPK.Optimizer}()
    @variable(model,  $\delta$ )
    @variable(model, b[i=1:m]  $\geq$  0)
    @constraint(model, sum(b) == 1.0)
    for a in  $\Gamma$ 
        @constraint(model, ( $\alpha$ -a)·b  $\geq$   $\delta$ )
    end
    @objective(model, Max,  $\delta$ )
    optimize!(model)
    return value( $\delta$ ) > 0 ? value.(b) : nothing
end

```

Algorithm 20.6. A method for finding the belief vector \mathbf{b} for which the alpha vector α improves the most compared to the set of alpha vectors Γ . Nothing is returned if no such belief exists. The packages JuMP.jl and GLPK.jl provide a mathematical optimization framework and a solver for linear programs, respectively.

Algorithm 20.7 provides a procedure that uses algorithm 20.6 for finding the dominating alpha vectors in a set Γ . Initially, all of the alpha vectors are candidates for being dominating. We then choose one of the candidates and determine the belief \mathbf{b} where the candidate leads to the greatest improvement in value compared to all other alpha vectors in the dominating set. If the candidate does not bring improvement, we remove it as a candidate. If it does bring improvement, we move an alpha vector from the candidate set that brings the greatest improvement

at b to the dominating set. The process continues until there are no longer any candidates. We can prune away any alpha vectors and associated conditional plans that are not dominating at any belief point. Example 20.3 demonstrates pruning on the crying baby problem.

```
function find_dominating( $\Gamma$ )
  n = length( $\Gamma$ )
  candidates, dominating = trues(n), falses(n)
  while any(candidates)
    i = findfirst(candidates)
    b = find_maximal_belief( $\Gamma$ [i],  $\Gamma$ [dominating])
    if b === nothing
      candidates[i] = false
    else
      k = argmax([candidates[j] ? b. $\Gamma$ [j] : -Inf for j in 1:n])
      candidates[k], dominating[k] = false, true
    end
  end
  return dominating
end

function prune(plans,  $\Gamma$ )
  d = find_dominating( $\Gamma$ )
  return (plans[d],  $\Gamma$ [d])
end
```

Algorithm 20.7. A method for pruning dominated alpha vectors and associated plans. The `find_dominating` function identifies all of the dominating alpha vectors in the set Γ . It uses binary vectors `candidates` and `dominating` to track which alpha vectors are candidates for inclusion in the dominating set and which are currently in the dominating set, respectively.


20.5 Value Iteration

The value iteration algorithm for MDPs can be adapted for POMDPs.⁶ POMDP *value iteration* (algorithm 20.8) begins by constructing all one-step plans. We prune any plans that are never optimal for any initial belief. Then, we expand all combinations of one-step plans to produce two-step plans. Again, we prune any suboptimal plans from consideration. This procedure of alternating between expansion and pruning is repeated until the desired horizon is reached. Figure 20.3 demonstrates value iteration on the crying baby problem.

⁶ This section describes a version of value iteration in terms of conditional plans and alpha vectors. For a version that only uses alpha vectors, see A. R. Cassandra, M. L. Littman, and N. L. Zhang, “Incremental Pruning: A Simple, Fast, Exact Method for Partially Observable Markov Decision Processes,” in *Conference on Uncertainty in Artificial Intelligence (UAI)*, 1997.

We can construct all two-step plans for the crying baby problem. There are $3^3 = 27$ such plans.

The expected utility for each plan in belief space is plotted below. We find that two plans dominate all others. These dominating plans are the only ones that need be considered as subplans for optimal three-step plans.



Example 20.3. The expected utility over the belief space for all two-step plans for the crying baby problem (appendix F.7). The thick lines are optimal for some beliefs, whereas the thin lines are dominated.

```
function value_iteration( $\mathcal{P}$ ::POMDP, k_max)
     $S, \mathcal{A}, R = \mathcal{P}.S, \mathcal{P}.A, \mathcal{P}.R$ 
    plans = [ConditionalPlan(a) for a in  $\mathcal{A}$ ]
     $\Gamma = [[R(s,a) \text{ for } s \text{ in } S] \text{ for } a \text{ in } \mathcal{A}]$ 
    plans,  $\Gamma$  = prune(plans,  $\Gamma$ )
    for k in 2:k_max
        plans,  $\Gamma$  = expand(plans,  $\Gamma$ ,  $\mathcal{P}$ )
        plans,  $\Gamma$  = prune(plans,  $\Gamma$ )
    end
    return (plans,  $\Gamma$ )
end

function solve( $M$ ::ValueIteration,  $\mathcal{P}$ ::POMDP)
    plans,  $\Gamma$  = value_iteration( $\mathcal{P}$ ,  $M.k\_max$ )
    return LookaheadAlphaVectorPolicy( $\mathcal{P}$ ,  $\Gamma$ )
end
```

Algorithm 20.8. Value iteration for POMDPs, which finds the dominating h -step plans for a finite-horizon POMDP of horizon k_max by iteratively constructing optimal subplans. The `ValueIteration` structure is the same as what was defined in algorithm 7.8 in the context of MDPs.




Figure 20.3. POMDP value iteration used to find the optimal value function for the crying baby problem to various horizons.

The expansion step (algorithm 20.9) in this process constructs all possible $k + 1$ -step plans from a set of k -step plans. New plans can be constructed using a new first action and all possible combinations of the k -step plans as subplans, as shown in figure 20.4. While plans can also be extended by adding additional actions to the ends of subplans, top-level expansion allows alpha vectors constructed for the k -step plans to be used to efficiently construct alpha vectors for the $(k + 1)$ -step plans.

Computing the alpha vector associated with a plan π from a set of alpha vectors associated with its subplans can be done as follows. We use α_o to represent the alpha vector associated with subplan $\pi(o)$. The alpha vector associated with π is then:

$$\alpha(s) = R(s, \pi()) + \gamma \sum_{s'} T(s' | s, \pi()) \sum_o O(o | \pi(), s') \alpha_o(s') \quad (20.17)$$

Even for relatively simple problems to shallow depths, computing alpha vectors from subplans in this way is much more efficient than computing them from scratch as in algorithm 20.2.

20.6 Linear Policies

As discussed in section 19.3, the belief state in a problem with linear-Gaussian dynamics can be represented by a Gaussian distribution $\mathcal{N}(\mu_b, \Sigma_b)$. If the reward function is quadratic, then it can be shown that the optimal policy can be computed exactly offline using a process that is often called *linear-quadratic-Gaussian* (LQG) control. The optimal action is obtained in an identical manner as in section 7.8, but the μ_b computed using the linear Gaussian filter is treated as the true state. With each observation, we simply use the filter to update our μ_b and obtain an optimal action by multiplying μ_b with the policy matrix from algorithm 7.11. Example 20.4 demonstrates this process.

20.7 Summary

- Exact solutions for POMDPs can typically only be obtained for finite-horizon discrete POMDPs.




Figure 20.4. A $k + 1$ step plan can be constructed using a new initial action leading to any combination of k -step subplans.

```

function ConditionalPlan( $\mathcal{P}$ ::POMDP, a, plans)
    subplans = Dict( $o \Rightarrow \pi$  for ( $o, \pi$ ) in zip( $\mathcal{P}.O$ , plans))
    return ConditionalPlan(a, subplans)
end

function combine_lookahead( $\mathcal{P}$ ::POMDP, s, a,  $\Gamma_o$ )
     $S, O, T, O, R, \gamma = \mathcal{P}.S, \mathcal{P}.O, \mathcal{P}.T, \mathcal{P}.O, \mathcal{P}.R, \mathcal{P}.\gamma$ 
     $U'(s', i) = \text{sum}(O(a, s', o) * \alpha[i] \text{ for } (o, \alpha) \text{ in zip}(O, \Gamma_o))$ 
    return  $R(s, a) + \gamma * \text{sum}(T(s, a, s') * U'(s', i) \text{ for } (i, s') \text{ in enumerate}(S))$ 
end

function combine_alphavector( $\mathcal{P}$ ::POMDP, a,  $\Gamma_o$ )
    return [combine_lookahead( $\mathcal{P}$ , s, a,  $\Gamma_o$ ) for s in  $\mathcal{P}.S$ ]
end

function expand(plans,  $\Gamma, \mathcal{P}$ )
     $S, A, O, T, O, R = \mathcal{P}.S, \mathcal{P}.A, \mathcal{P}.O, \mathcal{P}.T, \mathcal{P}.O, \mathcal{P}.R$ 
    plans',  $\Gamma' = [], []$ 
    for a in A
        # iterate over all possible mappings from observations to plans
        for inds in product([eachindex(plans) for o in O]...)
             $\pi_o = \text{plans}[[\text{inds}...]]$ 
             $\Gamma_o = \Gamma[[\text{inds}...]]$ 
             $\pi = \text{ConditionalPlan}(\mathcal{P}, a, \pi_o)$ 
             $\alpha = \text{combine_alphavector}(\mathcal{P}, a, \Gamma_o)$ 
            push!(plans',  $\pi$ )
            push!( $\Gamma'$ ,  $\alpha$ )
        end
    end
    return (plans',  $\Gamma'$ )
end

```

Algorithm 20.9. The expansion step in value iteration, which constructs all $k + 1$ -step conditional plans and associated alpha vectors from a set of k -step conditional plans and alpha vectors. The way we combine alpha vectors of subplans follows equation (20.17).

Consider a satellite navigating in two dimensions, neglecting gravity, drag, or other external forces. The satellite can use its thrusters to accelerate in any direction with linear dynamics:

$$\begin{bmatrix} x \\ y \\ \dot{x} \\ \dot{y} \end{bmatrix} \leftarrow \begin{bmatrix} 1 & 0 & \Delta t & 0 \\ 0 & 1 & 0 & \Delta t \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} x \\ y \\ \dot{x} \\ \dot{y} \end{bmatrix} + \begin{bmatrix} \frac{1}{2}\Delta t^2 & 0 \\ 0 & \frac{1}{2}\Delta t^2 \\ \Delta t & 0 \\ 0 & \Delta t \end{bmatrix} \begin{bmatrix} \ddot{x} \\ \ddot{y} \end{bmatrix} + \epsilon$$

where Δt is the duration of a time-step and ϵ is zero-mean Gaussian noise with covariance $\Delta t/20\mathbf{I}$.

We seek to place the satellite in its orbital slot at the origin, while minimizing fuel use. Our quadratic reward function is:


$$R(\mathbf{s}, \mathbf{a}) = -\mathbf{s}^\top \begin{bmatrix} \mathbf{I}_{2 \times 2} & \mathbf{0}_{2 \times 2} \\ \mathbf{0}_{2 \times 2} & \mathbf{0}_{2 \times 2} \end{bmatrix} \mathbf{s} - 2\mathbf{a}^\top \mathbf{a}$$

The satellite's sensors only estimate its position with the linear observation:

$$\mathbf{o} = \begin{bmatrix} \mathbf{I}_{2 \times 2} & \mathbf{0}_{2 \times 2} \end{bmatrix} \mathbf{s} + \varepsilon$$

where ε is zero-mean Gaussian noise with covariance $\Delta t/10\mathbf{I}$.

Below are 50 trajectories from 10-step rollouts using the optimal policy for $\Delta t = 1$ and a Kalman filter to track the belief. In each case, the satellite was started at $\mathbf{s} = \boldsymbol{\mu}_b = [-5, 2, 0, 1]$ with $\boldsymbol{\Sigma}_b = [\mathbf{I} \ 0; 0 \ 0.25\mathbf{I}]$.



Example 20.4. An example of an optimal policy used for a POMDP with linear Gaussian dynamics and quadratic reward.

- Policies for these problems can be represented as conditional plans, which are trees that describe the actions to take based on the observations.
- Alpha vectors contain the expected utility when starting from different states and following a particular conditional plan.
- Alpha vectors can also serve as an alternative representation of a POMDP policy.
- POMDP value iteration can avoid the computational burden of enumerating all conditional plans by iteratively computing subplans and pruning those that are suboptimal.
- Linear-Gaussian problems with quadratic reward can be solved exactly using methods very similar to those derived for the fully observable case.

20.8 Exercises

Exercise 20.1. Can every POMDP be framed as an MDP?

Solution: Yes. Any POMDP can equivalently be viewed as a belief-state MDP whose state space is the space of beliefs in the POMDP, whose action space is the same as that of the POMDP, and whose transition function is given by equation (20.1).

Exercise 20.2. What are the alpha vectors for the one-step crying baby problem (appendix F.7)? Are all actions dominant?

Solution: There are three one-step conditional plans, one for each action, resulting in three alpha vectors. The optimal one-step policy must choose between these actions given the current belief. The one-step alpha vectors for a POMDP can be obtained from the optimal one-step belief value function:

$$U^*(b) = \max_a \sum_s b(s) R(s, a)$$

Feeding the baby yields an expected reward:

$$\begin{aligned} R(\text{hungry}, \text{feed})P(\text{hungry}) + R(\text{sated}, \text{feed})P(\text{sated}) \\ = -15P(\text{hungry}) - 5(1 - P(\text{hungry})) \\ = -10P(\text{hungry}) - 5 \end{aligned}$$


Singing to the baby yields an expected reward:

$$\begin{aligned} & R(\text{hungry}, \text{sing})P(\text{hungry}) + R(\text{sated}, \text{sing})P(\text{sated}) \\ &= -10.5P(\text{hungry}) - 0.5(1 - P(\text{hungry})) \\ &= -10P(\text{hungry}) - 0.5 \end{aligned}$$

Ignoring the baby yields an expected reward:

$$\begin{aligned} & R(\text{hungry}, \text{ignore})P(\text{hungry}) + R(\text{sated}, \text{ignore})P(\text{sated}) \\ &= -10P(\text{hungry}) \end{aligned}$$

The expected reward for each action is plotted below over the belief space:



We find that under a one-step horizon it is never optimal to feed or sing to the baby. The ignore action is dominant.

Exercise 20.3. Why does the implementation of value iteration in algorithm 20.8 call expand plans in algorithm 20.9 rather than evaluating the plan in algorithm 20.2 to obtain alpha vectors for each new conditional plan?

Solution: The plan evaluation method applies equation (20.8) recursively to evaluate the expected utility for a conditional plan. Conditional plans grow very large as the horizon increases. POMDP value iteration can save computation by using the alpha vectors for the subplans from the previous iteration:

$$U^\pi(s) = R(s, \pi()) + \gamma \left[\sum_{s'} T(s' | s, \pi()) \sum_o O(o | \pi(), s') \alpha_{s'}^{\pi(o)} \right]$$

Exercise 20.4. Does the number of conditional plans increase faster with the number of actions or with the number of observations?

Solution: Recall that there are $|\mathcal{A}|^{(|\mathcal{O}|^h - 1)/(|\mathcal{O}| - 1)}$ possible h -step plans. Exponential growth (n^x) is faster than polynomial growth (x^n), and we have better-than exponential growth in $|\mathcal{O}|$ and polynomial growth in $|\mathcal{A}|$. The number of plans thus increases faster with respect to the number of observations. To demonstrate, let us use $|\mathcal{A}| = 3$, $|\mathcal{O}| = 3$, and $h = 3$ as a baseline. The baseline has 1,594,323 plans. Incrementing the number of actions results in 67,108,864 plans, whereas incrementing the number of observations results in 10,460,353,203 plans.

Exercise 20.5. Suppose we have a patient, and we are unsure whether or not they have a particular disease. We do have three diagnostic tests, each with different probabilities that they will correctly indicate whether or not the disease is present. While the patient is in our office, we have the option to administer potentially multiple diagnostic tests in sequence. We observe the outcome of each diagnostic test immediately. In addition, we can repeat any diagnostic test multiple times, with the outcomes of all tests being conditionally independent of each other given the presence or absence of the disease. When we are done with the tests, we decide whether to treat the disease or send the patient home without treatment. Explain how you would define the various components of a POMDP formulation.

Solution: We have three states:

1. s_{healthy} : the patient does not have the disease
2. s_{disease} : the patient has the disease
3. s_{terminal} : the interaction is over (terminal state)

We have five actions:

1. a_1 : administer test 1
2. a_2 : administer test 2
3. a_3 : administer test 3
4. a_{treat} : administer treatment and send patient home
5. a_{stop} : send patient home without treatment

We have three observations:

1. o_{healthy} : the outcome of the test (if administered) indicates the patient is healthy
2. o_{disease} : the outcome of the test (if administered) indicates the patient has the disease
3. o_{terminal} : a test was not administered

The transition model would be deterministic with:

$$T(s' | s, a) = \begin{cases} 1 & \text{if } a \in \{a_{\text{treat}}, a_{\text{stop}}\} \wedge s' = s_{\text{stop}} \\ 1 & \text{if } s = s' \\ 0 & \text{otherwise} \end{cases}$$

The reward function would be a function of the cost of administering treatment and each test as well as the cost of not treating the disease if it is indeed present. The reward available from s_{terminal} is 0. The observation model assigns probabilities to correct and incorrect observations of the disease state as a result of a diagnostic test from one of the nonterminal states. The initial belief would assign our prior probability to whether or not the patient has the disease, with zero probability assigned to the terminal state.

Exercise 20.6. Why might we want to perform the same test multiple times in the previous exercise?

Solution: Depending on the probability of incorrect results, we may want to perform the same test multiple times to improve our confidence in whether the patient has the disease or is healthy. The results of the tests are independent given the disease state.

Exercise 20.7. Suppose we have three alpha vectors: $[1, 0]$, $[0, 1]$, and $[\theta, \theta]$, for some constant θ . Under what conditions on θ can we prune alpha vectors?

Solution: We can prune alpha vectors if $\theta < 0.5$ or $\theta > 1$. If $\theta < 0.5$, then $[\theta, \theta]$ is dominated by the other two alpha vectors. If $\theta > 1$, then $[\theta, \theta]$ dominates the other two alpha vectors.

Exercise 20.8. We have $\Gamma = \{[1, 0], [0, 1]\}$ and $\alpha = [0.7, 0.7]$. What belief \mathbf{b} maximizes the utility gap δ as defined by the linear program in equation (20.16)?

Solution: The alpha vectors in Γ are shown in blue and the alpha vector α is shown in red. We care only about the region where $0.3 \leq b_2 \leq 0.7$ where α dominates the alpha vectors in Γ ; in other words, where the red line is above the blue lines. The point where the gap between the red line and the maximum of the blue lines occurs at $b_2 = 0.5$ with a gap of $\delta = 0.2$. Hence, the belief that maximizes this gap is $\mathbf{b} = [0.5, 0.5]$.

